Application Demonstration

Inverse of the Error Function

Univ.-Prof. Dr.-Ing. habil. J. BETTEN
RWTH University Aachen
Mathematical Models in Materials Science and Continuum Mechanics
Augustinerbach 4 - 22
D-52064 A a c h e n / Germany
betten@mmw.rwth-aachen.de

Abstract

Forming the inverse of the *GAUSS* error function erf(x) a best approximation to erf(x) has been discussed, the inverse of which can easily be determined. The deviation between the error function and its best approximation has been calculated by considering the L-two error norm.

Keywords: Best approximation; approximant; inverse; L-two error norm

Introduction

As has been explained by *BETTEN* (2005) in more detail the creep behavior of several materials can be interpreted as a *diffusion controlled process.* Calculating this process we need the inverse of the error function. The *GAUSS error function* is implied in the Maple software as a standard function. An approximation of its inverse is discussed in the following.

Best Approximation

An approximation to the error function erf(xi) on [0,r] is assumed by the *hyperbolic tangent:*

```
> restart:
> approximant:=tanh(a*xi);
```

$$approximant := \tanh(a\,\xi) \tag{2.1}$$

This function is suitable because it is similar to erf(xi). Furthermore, the inverse can easily be determined:

```
> inverse:=(1/a)*arctanh(xi);
```

$$inverse := \frac{\operatorname{arctanh}(\xi)}{a} \tag{2.2}$$

Note that the *Area Tangent*, artanh(...), is indicated as arctanh(...) in MAPLE. The *best* *approximation* by the *hyperbolic tangent* is guaranteed by an *optimal parameter* a which minimizes the L-two error norm:

```
> L_two[0..r]:=sqrt((1/r)*int((erf(xi)-tanh(a*xi))^2, xi=0..r))=
  minimum;
```

$$L\_two_{0..r} := \sqrt{\frac{1}{r}\left(\int_0^r \left(\mathrm{erf}(\xi) - \tanh(a\,\xi)\right)^2 \mathrm{d}\xi\right)} = minimum \tag{2.3}$$

Thus, the derivative of the integral should be equal to zero.

```
> derivative[0..r]:=diff(int((erf(xi)-tanh(a*xi))^2, xi=0..r),a)
  ;
```

$$derivative_{0..r} := \int_0^r \left(-2\left(\mathrm{erf}(\xi) - \tanh(a\,\xi)\right)\left(1 - \tanh(a\,\xi)^2\right)\xi\right) \mathrm{d}\xi \tag{2.4}$$

Depending on the range [0, r] considered, we obtain the following optimal parameters:

```
> for r in [1,2,3,4,5,infinity] do
  a[optimal][0..r]:= fsolve(int(xi*(erf(xi)-tanh(a_*xi))*(1-
  (tanh(a_*xi))^2), xi=0..r)=0,a_)
  od;
```

$$a_{optimal_{0..1}} := 1.172868316 \tag{2.5}$$

$$a_{optimal_{0..2}} := 1.201270935$$

$$a_{optimal_{0..3}} := 1.202760580$$

$$a_{optimal_{0..4}} := 1.202782281$$

$$a_{optimal_{0..5}} := 1.202782515$$

$$a_{optimal_{0..\infty}} := 1.202782517$$

The corresponding L-two error norms are given as:

```
> for r in [1,2,3,4,5,infinity] do
  L_two[0..r]:=evalf(sqrt((1/r)*int((erf(xi)-tanh(a[optimal][0..
  r]*xi))^2,xi=0..r)))
  od;
```

$$L\_two_{0..1} := 0.008219954058 \tag{2.6}$$

$$L\_two_{0..2} := 0.01428838030$$

$$L\_two_{0..3} := 0.01216051374$$

$$L\_two_{0..4} := 0.01053649880$$

$$L\_two_{0..5} := 0.009424169402$$

$$L\_two_{0..\infty} := 0.$$

Inverse Functions

Depending on the range [0, r] we obtain the following inverse functions:

```
> for r in [1,2,3,4,5,infinity] do
  inverse[0..r]:=(1/a[optimal][0..r])*arctanh(xi)
  od;
```

$$inverse_{0..1} := 0.8526106353\ \mathrm{arctanh}(\xi) \qquad (3.1)$$

$$inverse_{0..2} := 0.8324516734\ \mathrm{arctanh}(\xi)$$

$$inverse_{0..3} := 0.8314206640\ \mathrm{arctanh}(\xi)$$

$$inverse_{0..4} := 0.8314056632\ \mathrm{arctanh}(\xi)$$

$$inverse_{0..5} := 0.8314055014\ \mathrm{arctanh}(\xi)$$

$$inverse_{0..\infty} := 0.8314055001\ \mathrm{arctanh}(\xi)$$

Examples

In the following some examples should be plotted:

```
> alias(H=Heaviside,th=thickness):
> plot({1,H(xi-2),erf(xi),tanh(a[optimal][0..2]*xi)}, xi=0.
  .2.001,color=black);
```




```
> Delta(xi):=erf(xi)-tanh(a[optimal][0..2]*xi);
```

$$\Delta(\xi) := \mathrm{erf}(\xi) - \tanh(1.201270935\ \xi) \qquad (4.1)$$

```
> for i from 1 to 3 do
  zero[i-1]:=fsolve(Delta(xi)=0,xi,(i-1)/2..i/2)
  od;
```

(4.2)

$$zero_0 := 0. \quad (4.2)$$

$$zero_1 := 0.8439158081$$

$$zero_2 := fsolve\left(\text{erf}(\xi) - \tanh(1.201270935\,\xi) = 0, \xi, 1 .. \frac{3}{2}\right)$$

```
> plot1:=plot({-0.02,0.02,0.02*H(xi-2),-0.02*H(xi-2)}, xi=0.
  .2.001,color=black):
> plot2:=plot(Delta(xi),xi=0..2,color=black,th=3, title=
  "deviation between erf(xi) and tanh(a*xi) on [0, 2]"):
> plots[display]({plot1,plot2});
```




```
> L_two[0..2]:=sqrt((1/2)*Int((Delta)^2,xi=0..2))= evalf(sqrt(
  (1/2)*int((Delta(xi))^2,xi=0..2)));
```

$$L\_two_{0..2} := \frac{1}{2}\sqrt{2}\sqrt{\int_0^2 \Delta^2 \, d\xi} = 0.01428838030 \quad (4.3)$$

```
> for i in [1.18,1.19,a[optimal][0..2],1.21,1.22] do
  L_two[i]:=evalf(sqrt((1/2)*int((erf(xi)-tanh(i*xi))^2, xi=0.
  .2)))
  od;
```

$$L\_two_{1.18} := 0.01529807823 \quad (4.4)$$

$$L\_two_{1.19} := 0.01457543612$$

$$L\_two_{1.201270935} := 0.01428838030$$

$$L\_two_{1.21} := 0.01445712660$$

$$L\_two_{1.22} := 0.01504066478$$

For the value a =1.201270935 the L-two error norm is minimal in the range xi = [0, 2]. The following Figures illustrate the error function erf(xi) and some inverse approximations (1/a)* arctanh(xi):

```
> plot1:=plot({1,xi,H(xi-1)}, xi=0..1.001,scaling=constrained,
  color=black):
```

```
> plot2:=plot({erf(xi),(1/a[optimal][0..1])*arctanh(xi)}, xi=0.
  .1,0..1,scaling=constrained,color=black,th=2):
> plots[display]({plot1,plot2});
```




```
> plot1:=plot({1,2,xi,2*H(xi-2),2*H(xi-1),-2*H(xi-1.002)}, xi=0.
  .2.001,scaling=constrained,color=black):
> plot2:=plot({erf(xi),(1/a[optimal][0..2])*arctanh(xi)}, xi=0.
  .2,0..2,color=black,th=2):
> plots[display]({plot1,plot2});
```




```
> plot1:=plot({1,5,xi,5*H(xi-5),5*H(xi-1),-5*H(xi-1.002)}, xi=0.
  .5.001,scaling=constrained,color=black):
> plot2:=plot({erf(xi),(1/a[optimal][0..5])*arctanh(xi)}, xi=0.
  .5,0..5,color=black,th=2):
> plots[display]({plot1,plot2});
```



The results show that the approximant $\tanh(a \cdot xi)$ furnishes a suitable approximation to the error function $\operatorname{erf}(xi)$. In view of the inverse $\left(\frac{1}{a}\right)\operatorname{arctanh}(xi)$ and the approximant $\tanh(a \cdot xi)$ the influence of the range [0, r] on the parameter a is less important.

```
> delta[infinity,1]:=approximant[infinity]-approximant[1]= tanh
  (1.202782517*xi)-tanh(1.172868316*xi);
>
```

$$\delta_{\infty,1} := \tanh(a\,\xi)_{\infty} - \tanh(a\,\xi)_{1} = \tanh(1.202782517\,\xi) - \tanh(1.172868316\,\xi) \qquad \textbf{(4.5)}$$

```
> delta[infinity,2]:=approximant[infinity]-approximant[2]= tanh
  (1.202782517*xi)-tanh(1.201270935*xi);
```

$$\delta_{\infty,2} := \tanh(a\,\xi)_{\infty} - \tanh(a\,\xi)_{2} = \tanh(1.202782517\,\xi) - \tanh(1.201270935\,\xi) \qquad \textbf{(4.6)}$$

```
> plot1:=plot({H(xi-5),tanh(1.202782517*xi),tanh(1.172868316*xi)
  }, xi=0..5.001,color=black):
> plot2:=plot(1,xi=0..5,color=black, title="two approximants
  tanh(a*xi)"):
> plots[display]({plot1,plot2});
```



```
> plot1:= plot({0.012*H(xi-2),tanh(1.202782517*xi)-tanh
  (1.172868316*xi), tanh(1.202782517*xi)-tanh(1.201270935*xi)},
  xi=0..2.001,color=black):
> plot2:=plot(0.012,xi=0..2,color=black, title="distance delta
  [..] between approximants"):
> plots[display]({plot1,plot2});
```



The approximation dicussed above has been published in:
BETTEN, J.: **Creep Mechanics,** Springer-Verlag, Berlin/Heidelberg/New York, 2nd Edition 2005.